

SEÇİM SONUÇLARINI HESAPLAMA PROGRAMI

Burak ODER* / Bülent UYANIKER**

I- Programın Niteliği

Program, yaklaştırmalı, nisbî temsil sistemlerinde yaygın olarak kullanılan ve değişikliklerle yürürlükte olan 2839 sayılı Milletvekili Seçimi Kanunu'nda da benimsenen ¹ d'Hondt hesap usulü temelinde oluşturulmuştur. Ülke ve seçim çevresi barajlarının kullanılmasıyla bu temelde türev seçim sistemlerinin oluşturulması olanaklıdır. Ülke ve seçim çevresi barajları hesap usulü ise Milletvekili Kanunu'nda yapılan değişiklikler dikkate alınarak oluşturulmuştur.

Ülke barajı, program başında sabit bir oran olarak verilebilmektedir. Ülke barajının olmadığı haller için oran sıfır olarak verilmektedir.

Seçim çevresi barajı değişken (Kullanılan geçerli oy / milletvekili sayısı) veya sabit yüzdelik oran olarak verilebilmektedir. Değişken baraj tercihi, belirli sayıda milletvekili için alt sınırlamaya olanak da tanımaktadır. Bu alt sınırdan daha az milletvekili çıkaran seçim çevreleri için sabit bir yüzdelik oran verilebilmektedir.

Ülke barajı program başında bir kere uygulanarak barajı geçen partiler tesbit edilmekte; seçim çevresi barajı ise yine program başında bir kere verilmekte ama programdan çıkılana kadar her seçim çevresinde uygulanmaktadır.

Programın bu özellikleri 1983'ten bu yana uygulanan tüm seçim sistemlerine uygun hesap yapılmasına olanak tanımaktadır ².

* İ. Ü. İdare Hukuku Anabilim Dalı Araştırma Görevlisi.

** Hukukçu.

1. 34. madde 3. fıkra.

2. 3270 sayılı Kanunla ihdas edilen kontenjan milletvekilliğinin seçim çevresi barajı (kullanılan geçerli oy / [milletvekili sayısı - 1]) için, kontenjan milletvekili çıkaracak seçim çevrelerinde milletvekili sayısı bir eksiği ile verilerek baraj ve dağılım hesaplanabilir. 3403 sayılı Kanunla sadece, (kontenjan milletvekilliği de dahil olmak üzere) altı milletvekili çıkaracak seçim çevrelerine ilişkin olarak getirilen aynı nitelikli barajda ise, altı milletvekili çıkaracak seçim çevreleri için program yeniden çalıştırılarak hesap yapılabilir. 4125 sayılı Kanun ile ihdas edilen Türkiye milletvekilliği için tüm Türkiye tek seçim çevresi kabul edilerek dağılım hesaplanabilir.

II- Program Dili

Program, 5.0 versiyonundan itibaren tüm DOS işletim sistemlerinde ³ standard olarak bulunan QuickBasic (QBASIC) programlama dili ile yazılmıştır.

Bilgisayar kullanan, ancak programlama lisanını bilmeyenlere kolaylık sağlamak amacıyla satırlar numaralanmış, en az karmaşık komutlar kullanılmış ve ayrıntılı ekran düzenlemelerinden kaçınılmıştır. Programın algoritmasının anlaşılmasını sağlamak amacıyla değişkenlere kısaltmalarla ad verilmiş ve farklı kod sayfalarında (karakter setlerinde) sorun çıkmaması için Türkçe harfler bilinçli olarak kullanılmamıştır.

III- Programın Girilmesi

DOS "Prompt"unda ⁴ veya "DOS" dizisi altında ⁵ "QBASIC" komutunun verilmesiyle QBASIC programı çalışmaya başlayacaktır. Programın hata yapılmaksızın yazılması halinde ⁶ programı kaydederek ⁷ çalıştırmak olanaklıdır ⁸.

IV- Programın Çalışma Aşamaları

a. Kayıt dosyası

Program başında verilen dosya ismi ile açılan dosyada o çalışma oturumunda yapılan hesapların sonuçları saklanabilir. Bir dosya adı verilmemesi halinde program, her çalıştırmada içeriğini sileceği "out.dat" adlı bir dosya açmaktadır ⁹.

b. Parti adları

Parti adlarının kısaltma olarak girilmesi gerekmektedir. ENTER tuşuna basılması halinde seçime katılan parti sayısı ekrana yazılacaktır.

3. DOS 5.0, Windows 3.1 kullanıcılarının büyük çoğunluğunda üst işletim sistemi olarak bulunmaktadır.
4. Bilgisayarınız DOS'la açılıyorsa, ekranda büyük olasılıkla "C >" yazacaktır. Farklı bir biçimleme halinde bir harf (örn. D veya F) ve ">" işareti yazabilir. Bilgisayarınız Windows'la açılıyorsa Windows uygulamasına son vererek DOS'a çıkabilirsiniz.
5. "CD DOS" komutuyla dizine inebilirsiniz. Programa ve QBASIC programına son verdikten sonra "CD.." komutuyla tekrar DOS'a çıkabilirsiniz.
6. Programın yazımından sonra çalıştırılması ile hatalar tek tek ekrana gelecektir.
7. Bunun için menülerde yer alan SAVE (KAYDET) komutu kullanılır.
8. Bunun için menülerde yer alan RUN (İŞLET) komutu kullanılır.
9. Dosya içeriğini dosyayı QBASIC altında menülerdeki "OPEN (PROGRAM)" veya DOS altında "EDIT" komutu ile görebilirsiniz.

c. Ülke barajı

Ülke barajı yüzde işareti kullanılmadan bir sayı olarak (örneğin 10, 15, 22 vb.) verilmektedir. Ülke barajının olmaması halinde sıfır sayısı verilmelidir.

d. Tüm ülke geçerli oy sayısı ve parti ülke geneli toplam sayısı

Ülke barajı verilen hallerde program tüm ülke geçerli oy sayısını soracaktır. Bu sayı verilince ülke barajını ekranda gösterecek ve parti adlarını vererek ülke genelindeki oy sayılarını isteyecektir. Bu sayılar girildikten sonra, ülke barajını geçen partilerin adını ekrana yansıtacaktır.

Sayılar girilirken nokta veya virgöl gibi ayraçlar kullanılmamalıdır.

e. Seçim çevresi barajı

Program ilk olarak seçim çevresi barajı tipini soracaktır. Sabit oranda baraj oranının yüzde işareti kullanılmadan bir sayı olarak verilmesi gerekmektedir. Seçim çevresi barajı, ilgili seçim çevresinde kullanılan geçerli oy sayısı ile bu oran çarpılarak bulunmaktadır. Değişken oranın seçilmesi halinde baraj "seçim çevresinde kullanılan geçerli oy sayısı/milletvekili sayısı" formülü ile hesaplanacaktır. Program değişken oranın tüm seçim çevrelerinde mi, yoksa belirli sayıya kadar mı kullanılacağını, eğer bu ikincisi söz konusuysa daha küçük seçim çevrelerinde uygulanacak sabit oranı soracaktır. Örneğin 4125 sayılı Kanun, 2 ve 3 milletvekili çıkararak seçim çevrelerinde sabit bir oran (% 25) uygulanmasını öngörmektedir. Dolayısıyla değişken oranın alt sınırı 4 milletvekilidir. Programa böyle bir verinin girilmesi halinde 4 milletvekiline kadar formül, bunun altında sabit oran uygulanacaktır.

Çevre barajı seçilmişse, her seçim çevresinde gerekli verilerin girilmesinden sonra baraj hesaplanacak ve bu barajın altındaki partiler milletvekili dağıtımında dikkate alınmayacaktır. Tüm partilerin baraja takıldığı hallerde milletvekili dağıtımını baraj yokmuş gibi yapılacaktır.

f. Seçim çevreleri ve milletvekili dağıtımı

Program seçim çevresinin adını istemektedir. Ad yerine ENTER tuşuna basılması program için "dur" komutu anlamına gelmektedir.

Ad verildikten sonra program, baraj varsa geçerli oy ve milletvekili sayısını, yoksa sadece milletvekili sayısını isteyeceklerdir. Bu veriler girildikten sonra (varsa ülke barajını aşan) partilerin adını sırayla ekrana yansıtacak ve aldıkları

oyların girilmesini isteyecektir. Daha sonra işlem yaparak milletvekillerini d'Hondt usulüyle partilere dağıtacaktır. Bir seçim çevresinde 150 milletvekiline kadar hesap yapılabilmektedir.

g. İşlem sonuçlarını görmek, basmak veya programdan çıkmak

Seçim çevresi adı yerine ENTER tuşuna basılması halinde bu üç olasılık ekrana gelecektir. Yanlışlıkla tuşa basılması halinde barajlara ilişkin işlemlerin yeniden yapılmasına yol açmamak için seçim çevresine geri dönme olanağı eklenmiştir.

Sonuçları görme tercihi yapılırsa, herhangi bir karakter tuşuna basılarak seçim çevreleri ve ilgili veriler teker teker ekrana getirilebilir. Basma tercihi yapılırsa program DOS'ta yüklü yazıcıyı kullanarak sonuçları basacaktır¹⁰. Çıkış tercihi yapıldığında QBASIC'e geri dönecektir.

V- Programın EXE-File Haline Getirilebilmesi

Program EXE-File haline getirilebilir. Ancak, bu işlem için DOS altındaki QBASIC programı uygun değildir, tam versiyon bir QBASIC programının veya çevrim için hazırlanmış özel bir programın kullanılması gerekmektedir.

Program EXE-File haline getirilmeden önce 830'uncu satırdaki "STOP" komutu "SYSTEM" komutu ile değiştirilebilir. Bu halde program başka bir işleme gerek kalmaksızın kapanacak ve DOS'a dönecektir.

VI- Kullanım Çerçevesi

Programın akademik amaçlar dışında kullanımı, özellikle ticari amaçla yazılı veya başkaca bir yolla yeniden yayımı, dağıtımı veya kullanıcı programı haline getirilmesi geliştirenlerin yazılı onayını gerektirmektedir.

VII- Programla İlgili Sorun, Görüş ve Eleştiriler İçin

Programla ilgili sorun, görüş ve eleştirilerinizi "Burak Oder, Luxemburgerstr. 118/27, 50939 Köln, Almanya" posta adresine veya "a2786362 @ smail.nz.uni-koeln.de" e-mail adresine iletebilirsiniz.

10. Eğer DOS için yazıcı yüklenmemişse veya kullanılan yazıcı ile yüklü yazıcı farklıysa sonuçlar hiç basılamayabilir veya yanlış basılabilir. Böyle bir durumda sonuçları kaydettiğiniz dosyayı herhangi bir yazım programı ile açarak sonuçları basabilirsiniz.

```
10 CLS
20 il = 0: nparti = 0: saypar = 0
30 DIM top(3000, 2)
40 DIM iladi$(100)
50 DIM parti.mv(20, 100)
60 DIM parti.adi$(20)
70 DIM xparti.adi$(20)
80 DIM parti.oy(20, 100)
90 PRINT "Sonuclarin kaydedilecegi dosya adi veriniz,"
100 INPUT "vermezseniz sonuclar out.dat dosyasina yazilacaktir"; name$
110 IF name$ = "" THEN name$ = "out.dat"
120 OPEN name$ FOR OUTPUT AS 3
130 nparti = nparti + 1
140 PRINT nparti; ".";
150 LINE INPUT "Parti adi   : "; xparti.adi$(nparti)
160 IF xparti.adi$(nparti) <> "" GOTO 130
170 nparti = nparti - 1
180 np2 = nparti
190 PRINT "Secime"; nparti; " parti katilmistir."
200 INPUT "Ulke baraji %   : ", ubaraj
210 IF ubaraj = 0 THEN
220 FOR i = 1 TO nparti
230 parti.adi$(i) = xparti.adi$(i)
240 NEXT i
250 GOTO 550
260 END IF
270 INPUT "Tum ulke gecerli oy sayisi : ", ulkeoy
280 ubar.oy = ubaraj / 100! * ulkeoy
290 PRINT "Ulke baraji   : "; ubar.oy; "oydur."
```

```
300 FOR i = 1 TO nparti
310 PRINT UCASE$(LEFT$(xparti.adi$(i), 6)); TAB(8);
320 INPUT " Ulke geneli toplam oy sayisi : ", uparoy
330 IF uparoy > ubar.oy THEN
340 saypar = saypar + 1
350 parti.adi$(saypar) = xparti.adi$(i)
360 END IF
370 NEXT i
380 PRINT : PRINT
390 PRINT "Ulke barajini gecen partiler": PRINT
400 FOR i = 1 TO saypar
410 PRINT parti.adi$(i)
420 NEXT i
430 nparti = saypar
440 IF saypar = 0 THEN
450 PRINT "Hic bir parti ulke barajini gecememistir."
460 PRINT "Islem baraj sifir alinarak surduruluyor."
470 PRINT : PRINT
480 ubar.oy = 0
490 FOR i = 1 TO np2
500 saypar = saypar + 1
510 parti.adi$(saypar) = xparti.adi$(i)
520 NEXT i
530 nparti = saypar
540 END IF
550 PRINT : PRINT "Secim cevrelerine gecmek icin bir tusa basin ...";
560 z$ = INPUT$(1): CLS
570 PRINT "Cevre baraji tipi:": PRINT
```

```
580 PRINT "Cevre baraji yok [0]"
590 PRINT "Sabit oranli (orn. % 10) [1]"
600 PRINT "Degisken oranli (toplam oy/mv) [2]"
610 INPUT sec
620 baraj = 0: mvsay = 0
630 IF sec = 1 THEN
640 INPUT "Cevre baraji % : ", baraj
650 END IF
660 IF sec = 2 THEN
670 PRINT : PRINT "Toplam oy/milletvekili sayisi formulu tum secim"
680 PRINT "cevrelerinde gecerli ise [1], degilse gecerli oldugu"
690 INPUT "en az milletvekili sayisini giriniz: ", mvsay
700 IF mvsay > 1 THEN
710 INPUT "Bu sayinin altinda uygulanacak oran (orn. % 25) nedir: "; oran
720 END IF
730 END IF
740 CLS
750 PRINT
760 il = il + 1
770 INPUT "Secim cevresi : ", iladi$(il)
780 IF iladi$(il) = "" THEN
790 PRINT : PRINT " Sonuclari gormek icin [1]"
800 PRINT "Sonuclari basmak icin [2]"
810 PRINT "Secim cevrelerine donmek icin [3]"
820 INPUT "Cikmak icin [0] ", cik
830 IF cik = 0 THEN STOP
840 IF cik = 1 OR cik = 2 THEN
850 FOR k = 1 TO il - 1
860 PRINT iladi$(k)
```

```
870 IF cik = 2 THEN LPRINT iladi$(k)
880 FOR i = 1 TO nparti
890 IF cik = 1 THEN
900 PRINT parti.adi$(i), USING "###,###,### "; parti.oy(i, k + 1); parti.mv(i,
k + 1)
910 ELSE
920 LPRINT parti.adi$(i), USING "###,###,### "; parti.oy(i, k + 1);
parti.mv(i, k + 1)
930 END IF
940 NEXT i
950 IF cik = 1 THEN INPUT "Bir tusa basin ...", z$
960 NEXT k
970 CLS : GOTO 790
980 END IF
990 END IF
1000 IF cik = 3 THEN
1010 cik = - 1: il = 0: GOTO 750
1020 END IF
1030 IF cik > 3 THEN GOTO 790
1040 IF sec > 0 THEN INPUT "Toplam gecerli oy   :", toplam.oy
1050 IF toplam.oy < 0 THEN GOTO 1040
1060 INPUT "M. Vekili Sayisi   :", mv
1070 IF mv > 150 GOTO 1060
1080 PRINT
1090 IF sec = 1 THEN baraj.oy = toplam.oy / 100! * baraj: PRINT "BARAJ:";
baraj.oy
1100 IF sec = 2 THEN
1110 IF mv >= mvsay THEN
1120 baraj.oy = toplam.oy / mv: PRINT "BARAJ:"; baraj.oy
1130 ELSE
```



```
1140 baraj.oy = toplam.oy / 100! * oran: PRINT "BARAJ:"; baraj.oy
1150 END IF
1160 END IF
1170 FOR i = 1 TO nparti
1180 PRINT UCASE$(LEFT$(parti.adi$(i), 6)); TAB(8);
1190 INPUT " oy sayisi   :", parti.oy(i, il + 1)
1200 parti.oy(i, 1) = i
1210 NEXT i
1220 say = 0
1230 FOR i = 1 TO nparti
1240 IF parti.oy(i, il + 1) < baraj.oy GOTO 1300
1250 FOR k = 1 TO mv
1260 say = say + 1
1270 top(say, 1) = i
1280 top(say, 2) = parti.oy(i, il + 1) / k
1290 NEXT k
1300 NEXT i
1310 IF say = 0 THEN
1320 PRINT : PRINT "Hic bir parti cevre barajini gecememistir."
1330 PRINT "Baraj sifir alinarak devam ediliyor !"
1340 FOR i = 1 TO nparti
1350 FOR k = 1 TO mv
1360 say = say + 1
1370 top(say, 1) = i
1380 top(say, 2) = parti.oy(i, il + 1) / k
1390 NEXT k
1400 NEXT i
1410 END IF
1420 GOSUB 1550
```

```
1430 FOR i = 1 TO mv
1440 parti.mv(top(i, 1), il + 1) = parti.mv(top(i, 1), il + 1) + 1
1450 NEXT i
1460 PRINT #3, "----- "; iladi$(il); " -----"
1470 PRINT
1480 FOR i = 1 TO nparti
1490 PRINT UCASE$(parti.adi$(i)); TAB(8); "M. Vekili sayisi   : ";
1500 PRINT parti.mv(i, il + 1)
1510 PRINT #3, i; ". Parti   "; parti.adi$(i),
1520 PRINT #3, USING " ##,###,### "; parti.oy(i, il + 1); parti.mv(i, il + 1)
1530 NEXT i
1540 GOTO 750
1550 PRINT : PRINT "Milletvekili dagilimi hesaplaniyor..."
1560 FOR i = 1 TO say
1570 FOR j = i + 1 TO say
1580 IF top(j, 2) > top(i, 2) THEN
1590 c = top(j, 2)
1600 c2= top(j, 1)
1610 top(j, 1) = top(i, 1)
1620 top(i, 1) = c2
1630 top(j, 2) = top(i, 2)
1640 top(i,2) = c
1650 END IF
1660 NEXT j
1670 NEXT i
1680 RETURN
1690 END
```